Integral de Linha, circulação, Green

Aqui temos um procedimento para calcular automaticamente integrais de linha.

O procedimento para integral de linha

O procedimento **lineint** toma como entrada um campo vetorial F=[F1(x,y),F2(x,y)] em duas variaveis

x e y, mais uma curva parametrizada na forma xt, yt, trange
onde xt e yt sao as expressoes em t, e retorna como saida o valor da (circulação) integral de linha.

```
> lineint:=(F,xt,yt,trange)->
int(subs(x=xt,y=yt,F[1])*diff(xt,t)+
subs(x=xt,y=yt,F[2])*diff(yt,t),trange);
```

$$lineint := (F, xt, yt, trange) \rightarrow \int \mathrm{subs}(x = xt, y = yt, F_1) \left(\frac{\partial}{\partial t} xt\right) + \mathrm{subs}(x = xt, y = yt, F_2) \left(\frac{\partial}{\partial t} yt\right) dtran$$

O procedimento **lineintline** aplica para o caso especial, mas muito comum de curvas que são segmentos de reta ligando pontos extremos p e q . Ele toma como entrada um campo de vetores F=[F1 (x,y),F2(x,y)] em duas variaveis x e y, mais dois pontos p=[p1,p2] e q=[q1,q2] da curva e retorna o valor da (circulação) integral de linha.

```
> lineintline:=proc(F,p,q)
local vxt, vyt, xt, yt;
vxt:=q[1]-p[1];
vyt:=q[2]-p[2];
xt:=p[1]+t*vxt;
yt:=p[2]+t*vyt;
int(subs(x=xt,y=yt,F[1])*vxt+
subs(x=xt,y=yt,F[2])*vyt,t=0..1);
end:
```

Fluxos

O procedimento **fluxint** toma como entrada um campo de vetores F=[F1(x,y),F2(x,y)] em duas variáveis x e y , mais uma curva parametrizada na forma xt, yt, trange, onde xt e yt são as expressões em t, e retorna como saída o

valor (o fluxo) da integral de linha ..

```
> fluxint:=(F,xt,yt,trange)->
int(subs(x=xt,y=yt,F[1])*diff(yt,t)+
subs(x=xt,y=yt,F[2])*(-diff(xt,t)),trange);
```

$$fluxint := (F, xt, yt, trange) \rightarrow \int \mathrm{subs}(x = xt, y = yt, F_1)\left(\frac{\partial}{\partial t} yt\right) - \mathrm{subs}(x = xt, y = yt, F_2)\left(\frac{\partial}{\partial t} xt\right) dtra$$

O procedimento **fluxintline** se aplica para o caso especial, mas muito comum em que a curva é um segmento de reta definida pelos extremos pontos p e q. Ele toma como entrada um campo de vetores F=[F1(x,y),F2(x,y)] nas variáveis x e y, mais os pontos extremos p=[p1,p2] e q=[q1,q2] da curva, e retorna o valor da integral de linha (o fluxo).

```
> fluxintline:=proc(F,p,q)
local vxt, vyt, xt, yt;
vxt:=q[1]-p[1];
vyt:=q[2]-p[2];
xt:=p[1]+t*vxt;
yt:=p[2]+t*vyt;
int(subs(x=xt,y=yt,F[1])*vyt+
subs(x=xt,y=yt,F[2])*(-vxt),t=0..1);
end:
```

O Teorema de Green

Primeiro considere o caso linear e regiões simples (retângulos com lados paralelos aos eixos coordenados)

```
> linear_field:=[a*x+b*y,c*x+d*y];
```

$$linear\_field := [a\, x + b\, y, c\, x + d\, y]$$

Suponha que o retangulo está centrado em (x0,y0) com lados de comprimento 2dx
e 2dy paralelos aos eixos coordenados.

A circulação- integral de linha

```
> bott:=lineintline([a*x+b*y,c*x+d*y],[x0-dx,y0-dy],[x0+dx,y0-dy]);
```

```
> top:=lineintline([a*x+b*y,c*x+d*y],[x0+dx,y0+dy],[x0-dx,y0+dy]);
```

$$bott := 2\, dx\, a\, x0 + 2\, dx\, b\, y0 - 2\, dx\, b\, dy$$

$$top := -2\, dx\, a\, x0 - 2\, dx\, b\, y0 - 2\, dx\, b\, dy$$

```
> bott+top;
```

$$-4\,dx\,b\,dy$$

```
> rite:=lineintline([a*x+b*y,c*x+d*y],[x0+dx,y0-dy],[x0+dx,y0+dy]);
> lleft:=lineintline([a*x+b*y,c*x+d*y],[x0-dx,y0+dy],[x0-dx,y0-dy]);
```

$$rite := 2\,dy\,c\,x0 + 2\,dy\,c\,dx + 2\,dy\,d\,y0$$

$$lleft := -2\,dy\,c\,x0 + 2\,dy\,c\,dx - 2\,dy\,d\,y0$$

```
> rite+lleft;
```

$$4\,dy\,c\,dx$$

```
> bott+top+rite+lleft;
```

$$-4\,dx\,b\,dy + 4\,dy\,c\,dx$$

```
> simplify(");
```

$$-4\,dx\,b\,dy + 4\,dy\,c\,dx$$

O fluxo - integral de linha

```
> bott:=fluxintline([a*x+b*y,c*x+d*y],[x0-dx,y0-dy],[x0+dx,y0-dy]);
> top:=fluxintline([a*x+b*y,c*x+d*y],[x0+dx,y0+dy],[x0-dx,y0+dy]);
```

$$bott := -2\,dx\,c\,x0 - 2\,dx\,d\,y0 + 2\,dx\,d\,dy$$

$$top := 2\,dx\,c\,x0 + 2\,dx\,d\,y0 + 2\,dx\,d\,dy$$

```
> bott+top;
```

$$4\,dx\,d\,dy$$

```
> rite:=fluxintline([a*x+b*y,c*x+d*y],[x0+dx,y0-dy],[x0+dx,y0+dy]);
> lleft:=fluxintline([a*x+b*y,c*x+d*y],[x0-dx,y0+dy],[x0-dx,y0-dy]);
```

$$rite := 2\,dy\,a\,x0 + 2\,dy\,a\,dx + 2\,dy\,b\,y0$$

$$lleft := -2\,dy\,a\,x0 + 2\,dy\,a\,dx - 2\,dy\,b\,y0$$

```
> rite+lleft;
```

$$4\,dy\,a\,dx$$

```
> bott+top+rite+lleft;
```

$$4\,dx\,d\,dy + 4\,dy\,a\,dx$$

> **simplify(");**

$$4\,dx\,d\,dy + 4\,dy\,a\,dx$$

>

>